
ERWIN KALVELAGEN

Abstract. This document describes an implementation of Benders Decom-

position for solving two-stage Stochastic Linear Programming problems using
GAMS.

1. Introduction

The two-stage stochastic linear programming problem can be stated as [2, 3, 4]:

SLP minimize
x

cTx+ EωQ(x, ω)

Ax = b
x ≥ 0

where

Q(x, ω) = min
y
dTωy

Tωx+Wωy = hω

y ≥ 0

(1)

Here Eω is the expectation, and ω denotes a scenario or possible outcome with
respect to the probability space (Ω, P ). We will consider discrete distributions P
only, so we can write:

(2) EωQ(x, ω) =
∑
ω∈Ω

p(ω)Q(x, ω)

Using this we can formulate a large LP that forms the deterministic equivalent
problem:

min cTx+
∑
ω

p(ω)dTωyω

Ax = b

Tωx+Wωyω = hω

x ≥ 0, yω ≥ 0

(3)

Date: 17 januari 2003.

1



2 ERWIN KALVELAGEN

The structure of equation Tωx+Wωyω = hω is called L-shaped, which can be made
visible by writing it out:

(4)

T1x +W1y1 = h1

T2x +W2y2 = h2

T3x +W3y3 = h3

...
TKx +WKyK = hK

2. Benders Algorithm

The Benders’ algorithm[1, 7] to solve this problem can be formulated as follows
(we largely follow the notation in[6]).

Step 1: Initialization
ν := 1 {Iteration number}
UB :=∞ {Upper bound}
LB := −∞ {Lower bound}
Solve initial master problem:

min cTx
Ax = b

x ≥ 0
(5)

xν := x∗ {Optimal values}

Step 2: Sub problems
for ω ∈ Ω do

Solve the sub problem:

min dTωyω
Wωyω = hω − Tωxν

yω ≥ 0
(6)

yνω := y∗ω {Optimal values}
πνω := π∗ω {Optimal dual values}

end for
UB := min{UB, cTxν +

∑
ω∈Ω pωd

T
ωy

ν
ω}

Step 3: Convergence test
if (UB − LB)/(1 + LB) ≤ TOL then

Stop: required accuracy achieved
Return xν

end if

Step 4: Master problem

Solve the Master problem:

min cTx+ θ

Ax = b

θ ≥
∑
ω∈Ω

pω
(
−π`ω[Tωx+Wωy

`
ω − hω]

)
, ` = 1, . . . , ν − 1

x ≥ 0

(7)

xν := x∗ {Optimal values}
θ
ν

:= θ∗

LB := cTxν + θ
ν

go to step 2.
If the subproblems are infeasible, we need to include slightly different formulated

cuts. The optimality cuts shown here assume the subproblems could be solved to
an optimal feasible solution.

3. Example

The standard transportation model can be stated as:

TRANSPORT minimize
x

∑
i,j

ci,jxi,j∑
j

xi,j = si ∀i∑
i

xi,j = dj ∀j

xi,j ≥ 0

where ci,j are the unit transportation costs, si is the supply at plant i and dj is the
demand at location j.

Now consider the case that demand dj is stochastic. We assume that products
are shipped before the actual demand is observed. After the products arrive at
each location j, actual demand is known. If demand is not met, the sales is lost.
If the shipment is larger than the final demand, then the remaining products need
to be disposed of as they have no shelf life (i.e. they spoil). Unit disposal costs are
known.

To model this, first assume we somehow know the demand. This means we can
build a non-stochastic version of the model, also called the core model:

max
∑
j

pjSalesj −
∑
i,j

ci,jShipi,j −
∑
i

ciProdi −
∑
j

cjWastej

Prodi =
∑
j

Shipi,j

Prodi ≤ capi∑
i

Shipi,j = Salesj + Wastej

Salesj ≤ demandj
Salesj ≥ 0,Shipi,j ≥ 0,Prodi ≥ 0,Wastej ≥ 0

(8)

In this model we try to maximize profit, i.e. revenue minus costs, where costs
can be broken down in production costs, transportation costs and waste disposal
costs. This model is just a little bit more complex than the transportation model.

When we make demandj stochastic, the problem becomes more involved. We
assume there are three scenarios: pessimistic, average and optimistic. The deter-
ministic equivalent of the stochastic model could look like:

max Profit =
∑
j,ω

pjprobωSalesj,ω −
∑
i,j

ci,jShipi,j

−
∑
i

ciProdi −
∑
j,ω

cjprobωWastej,ω

Prod i =
∑
j

Shipi,j

Prod i ≤ capi∑
i

Shipi,j = Salesj,ω + Wastej,ω

Salesj,ω ≤ demandj
Salesj,ω ≥ 0,Shipi,j ≥ 0,Prodi ≥ 0,Wastej,ω ≥ 0

(9)

The following model implements the core model and the deterministic equivalent
formulation in GAMS:

3.0.1. Model stochlp.gms. 1

$ontext

Transportation problem with stochastic demands.
Simple three scenario problem.

Erwin Kalvelagen, January 2003

$offtext

sets
i ’factories’ /f1*f3/
j ’distribution centers’ /d1*d5/
s ’scenarios’ /lo,mid,hi/

;

parameter capacity(i) /f1 500, f2 450, f3 650/;

table demand(j,s) ’possible outcomes for demand’
lo mid hi

d1 150 160 170
d2 100 120 135
d3 250 270 300
d4 300 325 350
d5 600 700 800

;

parameter prob(s) ’probabilities’ /
lo 0.25
mid 0.5
hi 0.25

/;

table transcost(i,j) ’unit transportation cost’

1http://www.amsterdamoptimization.com/models/stochbenders/stochlp.gms

http://www.amsterdamoptimization.com/models/stochbenders/stochlp.gms

d1 d2 d3 d4 d5
f1 2.49 5.21 3.76 4.85 2.07
f2 1.46 2.54 1.83 1.86 4.76
f3 3.26 3.08 2.60 3.76 4.45

;

scalar prodcost ’unit production cost’ /14/;
scalar price ’sales price’ /24/;
scalar wastecost ’cost of removal of overstocked products’ /4/;

*-----------------------------------------------------------------------
* first we formulate a non-stochastic version of the model
* we just use ’mid’ values for the demand.
*-----------------------------------------------------------------------

variables
ship(i,j) ’shipments’
product(i) ’production’
sales(j) ’sales (actually sold)’
waste(j) ’overstocked products’
profit

;
positive variables ship,product,sales,waste;

equations
obj
production(i)
selling(j)

;

obj.. profit =e= sum(j, price*sales(j)) - sum((i,j), transcost(i,j)*ship(i,j))
- sum(j, wastecost*waste(j)) - sum(i,prodcost*product(i));

production(i).. product(i) =e= sum(j, ship(i,j));
product.up(i) = capacity(i);

selling(j).. sum(i, ship(i,j)) =e= sales(j)+waste(j);
sales.up(j) = demand(j,’mid’);

model nonstoch /obj,production,selling/;
solve nonstoch maximizing profit using lp;

display ship.l,product.l,sales.l,waste.l;

*-----------------------------------------------------------------------
* now we formulate a stochastic version of the model
* We form here the deterministic equivalent
*-----------------------------------------------------------------------

variables
salesw(j,s) ’stochastic version of sales’
wastew(j,s) ’stochastic version of waste’
received(j) ’amount of product received in distribution center’

;
positive variable salesw,wastew;

equations
objw
sellingw(j,s)
receive(j)

;

objw.. profit =e= sum((j,s),price*prob(s)*salesw(j,s))
- sum((i,j), transcost(i,j)*ship(i,j))
- sum((j,s), wastecost*prob(s)*wastew(j,s))
- sum(i,prodcost*product(i));

receive(j).. received(j) =e= sum(i, ship(i,j));

sellingw(j,s).. received(j) =e= salesw(j,s)+wastew(j,s);
salesw.up(j,s) = demand(j,s);

model stoch /objw,production,receive,sellingw/;
solve stoch maximizing profit using lp;

display ship.l,product.l,salesw.l,wastew.l,profit.l;

The optimal levels for the stage one variables from the deterministic model are:
---- 113 VARIABLE ship.L shipments

d1 d2 d3 d4 d5

f1 500.000
f2 150.000 300.000
f3 100.000 270.000 100.000

4. Benders formulation

This model is quite easily to solve using Benders’ Decomposition in GAMS. A
complete formulation is given below.

4.0.2. Model stochbenders.gms. 2

$ontext

Benders decomposition applied to a two-stage
stochastic linear programming problem.

Erwin Kalvelagen, januari 2003

$offtext

sets
i ’factories’ /f1*f3/
j ’distribution centers’ /d1*d5/
s ’scenarios’ /lo,mid,hi/

;

parameter capacity(i) /f1 500, f2 450, f3 650/;

table demand(j,s) ’possible outcomes for demand’
lo mid hi

d1 150 160 170
d2 100 120 135
d3 250 270 300
d4 300 325 350
d5 600 700 800

;

parameter prob(s) ’probabilities’ /
lo 0.25
mid 0.5
hi 0.25

/;

table transcost(i,j) ’unit transportation cost’
d1 d2 d3 d4 d5

f1 2.49 5.21 3.76 4.85 2.07
f2 1.46 2.54 1.83 1.86 4.76
f3 3.26 3.08 2.60 3.76 4.45

;

2http://www.amsterdamoptimization.com/models/stochbenders/stochbenders.gms

http://www.amsterdamoptimization.com/models/stochbenders/stochbenders.gms

scalar prodcost ’unit production cost’ /14/;
scalar price ’sales price’ /24/;
scalar wastecost ’cost of removal of overstocked products’ /4/;

*-----------------------------------------------------------------------
* Form the Benders master problem
*-----------------------------------------------------------------------

set
iter ’max Benders iterations’ /iter1*iter25/
dyniter(iter) ’dynamic subset’

;

positive variables
ship(i,j) ’shipments’
product(i) ’production’
slackproduct(i) ’slack’
received(j) ’quantity sent to market’

;
free variables

zmaster ’objective variable of master problem’
theta ’extra term in master obj’

;

equations
masterobj ’master objective function’
production(i) ’calculate production in each factory’
receive(j) ’calculate quantity to be send to markets’
prodcap(i) ’production capacity’
optcut(iter) ’Benders optimality cuts’

;

parameter
cutconst(iter) ’constants in optimality cuts’
cutcoeff(iter,j) ’coefficients in optimality cuts’

;

masterobj..
zmaster =e= sum((i,j), transcost(i,j)*ship(i,j))

+ sum(i,prodcost*product(i)) + theta;

receive(j).. received(j) =e= sum(i, ship(i,j));
production(i).. product(i) =e= sum(j, ship(i,j));
prodcap(i).. product(i) + slackproduct(i) =e= capacity(i);
optcut(dyniter).. theta =g= cutconst(dyniter) +

sum(j, cutcoeff(dyniter,j)*received(j));

model masterproblem /masterobj, receive, production, prodcap, optcut/;

*-----------------------------------------------------------------------
* Form the Benders’ subproblem
* Notice in equation selling we use the level value received.l, i.e.
* this is a constant
*-----------------------------------------------------------------------

positive variables
sales(j) ’sales (actually sold)’
waste(j) ’overstocked products’
slacksales(j) ’slack’

;
free variables

zsub ’objective variable of sub problem’
;

equations
subobj ’subproblem objective function’
selling(j) ’part of received is sold’
selmax(j) ’upperbound on sales’

;

parameter demnd(j) ’demand’;

subobj..
zsub =e= -sum(j, price*sales(j)) + sum(j, wastecost*waste(j));

selling(j).. sales(j) + waste(j) =e= received.l(j);
selmax(j).. sales(j) + slacksales(j) =e= demnd(j);

model subproblem /subobj,selling,selmax/;

*-------------------------------------------------------------------
* solver options
*-------------------------------------------------------------------

option limrow = 0;
option limcol = 0;
subproblem.solprint = 2;
masterproblem.solprint = 2;
subproblem.solvelink = 2;
masterproblem.solvelink = 2;

*-------------------------------------------------------------------
* Benders algorithm
*-------------------------------------------------------------------

*
* step 1: solve master without cuts
*

display "-----------------------------------------------------------------",
"Master without cuts",
"-----------------------------------------------------------------";

dyniter(iter) = NO;
cutconst(iter) = 0;
cutcoeff(iter,j) = 0;
theta.fx = 0;
solve masterproblem minimizing zmaster using lp;
display zmaster.l;

*
* repair bounds
*
theta.lo = -INF;
theta.up = INF;

scalar lowerbound /-INF/;
scalar upperbound /INF/;
parameter objsub(s);
scalar objmaster;
objmaster = zmaster.l;

scalar iteration;
scalar done /0/;

loop(iter$(not done),

iteration = ord(iter);
display "-----------------------------------------------------------------",

iteration,
"-----------------------------------------------------------------";

*
* solve subproblems
*

dyniter(iter) = yes;
loop(s,

demnd(j) = demand(j,s);

solve subproblem minimizing zsub using lp;
objsub(s) = zsub.l;
cutconst(iter) = cutconst(iter) - prob(s)*sum(j,(-selmax.m(j))*demand(j,s));
cutcoeff(iter,j) = cutcoeff(iter,j) - prob(s)*(-selling.m(j));

);

upperbound = min(upperbound, objmaster + sum(s, prob(s)*objsub(s)));

*
* convergence test
*

display lowerbound,upperbound;
if( (upperbound-lowerbound) < 0.001*(1+abs(lowerbound)),

display "Converged";
done = 1;

else

*
* solve masterproblem
*

solve masterproblem minimizing zmaster using lp;
display ship.l;

lowerbound = zmaster.l;
objmaster = zmaster.l-theta.l;

);

);

abort$(not done) "Too many iterations";

display "-----------------------------------------------------------------",
"Optimal stage one solution",
"-----------------------------------------------------------------",
ship.l;

The optimal levels for the stage one variables from this formulation are:
---- 221 VARIABLE ship.L shipments

d1 d2 d3 d4 d5

f1 500.000
f2 150.000 300.000
f3 100.000 270.000 100.000

5. Dual subproblem

In the model above we used a primal subproblem and its marginals (duals) in
the calculation of the cuts. We can also solve a dual subproblem directly. Such a
problem can be formulated as:
*-------------------------------------------------------------------
* Dual subproblem
*-------------------------------------------------------------------

variables
pi_selling(j) ’dual of equation selling’
pi_selmax(j) ’dual of equation selmax’

;

equations
dualobj
dualcon(j)

;

dualobj..
zsub =e= sum(j, received.l(j)*pi_selling(j)) + sum(j, demnd(j)*pi_selmax(j));

dualcon(j).. pi_selling(j) + pi_selmax(j) =l= -price;

pi_selling.up(j) = wastecost;
pi_selmax.up(j) = 0;

model dualsubproblem /dualobj,dualcon/;

The subproblem solution step now becomes:
*
* solve subproblems
*

dyniter(iter) = yes;
loop(s,

demnd(j) = demand(j,s);
solve dualsubproblem maximizing zsub using lp;
objsub(s) = zsub.l;
cutconst(iter) = cutconst(iter) - prob(s)*sum(j,(-pi_selmax.l(j))*demand(j,s));
cutcoeff(iter,j) = cutcoeff(iter,j) - prob(s)*(-pi_selling.l(j));

);

This problem is rather simple and can even be solved directly without invoking
an LP solver:

loop(j,
if (received.l(j)>demnd(j),

pselling(j) = wastecost;
pselmax(j) = -price-pselling(j);

else
pselling(j) = -price;
pselmax(j) = 0;

);
);

A slightly different variant of this problem is described in [5]: there the proba-
bilities are considered independent leading to 243 scenarios. This paper also imple-
ments the last suggestion, to solve the subproblems directly inside GAMS instead
of as using an LP problem.

References

1. J. F. Benders, Partitioning procedures for solving mixed-variables programming problems, Nu-

merische Mathematik 4 (1962), 238–252.
2. John R. Birge and François Louveaux, Introduction to Stochastic Programming, Springer Series

in Operations Research, Springer, 1997.

3. Gerd Infanger, Planning Under Uncertainty – Solving Large-Scale Stochastic Linear Programs,
Boyd & Fraser, 1994.

4. Peter Kall and Stein W. Wallace, Stochastic Programming, Wiley, 1994.

5. Erwin Kalvelagen, Two Stage Stochastic Linear Programming with GAMS, http://www.

amsterdamoptimization.com/pdf/twostage.pdf.

6. S. S. Nielsen and S.A. Zenios, Scalable Parallel Benders Decomposition for Stochastic Linear

Programming, Parallel Computing 23 (1997), 1069–1088.
7. R. M. van Slyke and R. J.-B. Wets, L-Shaped Linear Programs with Applications to Optimal

Control and Stochastic Programming, SIAM Journal on Applied Mathematics 17 (1969), 638–

663.

Amsterdam Optimization Modeling Group LLC, Washington D.C./The Hague
E-mail address: erwin@amsterdamoptimization.com

http://www.amsterdamoptimization.com/pdf/twostage.pdf
http://www.amsterdamoptimization.com/pdf/twostage.pdf

	1. Introduction
	2. Benders Algorithm
	3. Example
	4. Benders formulation
	5. Dual subproblem
	References

